A super-d number is a positive, decimal (base ten) integer   n   such that   d × nd   has at least   d   consecutive digits   d   where

   2 ≤ d ≤ 9

For instance, 753 is a super-3 number because 3 × 7533 = 1280873331.


Super-d   numbers are also shown on   MathWorld™   as   super-d    or   super-d.



Task

 Write a function/procedure/routine to find super-d numbers.
 For   d=2   through   d=6,   use the routine to show the first   10   super-d numbers.


Extra credit

 Show the first   10   super-7, super-8, and/or super-9 numbers   (optional).


See also

   Wolfram MathWorld - Super-d Number.
   OEIS: A014569 - Super-3 Numbers.


/*REXX program computes and displays the first  N  super─d  numbers for D from LO to HI.*/
numeric digits 100                               /*ensure enough decimal digs for calc. */
parse arg n LO HI .                              /*obtain optional arguments from the CL*/
if  n=='' |  n==","  then  n= 10                 /*the number of super─d numbers to calc*/
if LO=='' | LO==","  then LO=  2                 /*low  end of  D  for the super─d nums.*/
if HI=='' | HI==","  then HI=  9                 /*high  "   "  "   "   "     "      "  */
                                                 /* [↓]   process  D  from  LO ──►  HI. */
     do d=LO  to HI;     #= 0;     $=            /*count & list of super─d nums (so far)*/
     z= copies(d, d)                             /*the string that is being searched for*/
           do j=2  until #==n                    /*search for super─d numbers 'til found*/
           if pos(z, d * j**d)==0  then iterate  /*does product have the required reps? */
           #= # + 1;               $= $ j        /*bump counter;  add the number to list*/
           end   /*j*/
     say
     say center(' the first '     n     " super-"d 'numbers ',  digits(),  "═")
     say $
     end   /*d*/                                 /*stick a fork in it,  we're all done. */